Ecuaciones y sistemas de ecuaciones en Maple

José Luis Torres Rodríguez*

Febrero 2011

Maple proporciona al usuario un conjunto de funciones para manipulación y solución de ecuaciones y sistemas de ecuaciones, tanto algebraicas como ecuaciones más complicadas (por ejemplo, aquellas que contienen funciones trascendentales).

1. Definición de una ecuación

Una ecuación en Maple puede ser definida de la siguiente forma:

expresión1 = expresión2;

Generalmente es más práctico tener una ecuación asignada a una variable, esta asignación podemos hacerla de la siguiente manera:

nombre := ecuación;

Por ejemplo, definiremos una ecuación de segundo grado:

```
> ec1 := x^2 + 4*x + 5 = 1;
```

$$ec1 := x^2 + 4\,x + 5 = 1$$

2. Solución de una ecuación

Este sistema proporciona diversas funciones para la obtención de soluciones exactas y aproximadas de ecuaciones y sistemas de ecuaciones. A continuación haremos una revisión de estas dos formas de obtener las soluciones.

2.1. Obtención de soluciones exactas

Maple tiene la capacidad de resolver simbolicamente ecuaciones de segundo, tercero y cuarto grado en general, incluso casos particulares de ecuaciones polinomiales de orden mayor que cuatro y algunas ecuaciones trascendentales. Una de las instrucciones con la cual podemos resolver ecuaciones es **solve**, su sintaxis es:

solve(ecuación, var);

Donde **var** es la variable con respecto a la cual se desea resolver. Por ejemplo, definamos la siguiente ecuación:

```
> ec := 2*x = 9;
```

$$ec := 2\,x = 9$$

Resolveremos **ec** con respecto a la variable **x**:

```
> solve(ec, x);
```

*Coordinación de Cómputo, Facultad de Ciencias, UNAM

$$\frac{9}{2}$$

De la misma forma podemos resolver ecuaciones más complicadas, por ejemplo:

```
> ec2 := 2*x^3 - 8*x^2 + 3*x + 8 = 4*x^3 - 2*x^2;
```

$$ec2 := 2\,x^3 - 8\,x^2 + 3\,x + 8 = 4\,x^3 - 2\,x^2$$

```
> solve(ec2, x);
```

$$\frac{(2+2\,I\,\sqrt{53})^{(1/3)}}{2} + \frac{3}{(2+2\,I\,\sqrt{53})^{(1/3)}} - 1, -\frac{(2+2\,I\,\sqrt{53})^{(1/3)}}{4} - \frac{3}{2\,(2+2\,I\,\sqrt{53})^{(1/3)}} - 1$$
$$+\frac{1}{2}\,I\,\sqrt{3}\left(\frac{(2+2\,I\,\sqrt{53})^{(1/3)}}{2} - \frac{3}{(2+2\,I\,\sqrt{53})^{(1/3)}}\right), -\frac{(2+2\,I\,\sqrt{53})^{(1/3)}}{4}$$
$$-\frac{3}{2\,(2+2\,I\,\sqrt{53})^{(1/3)}} - 1 - \frac{1}{2}\,I\,\sqrt{3}\left(\frac{(2+2\,I\,\sqrt{53})^{(1/3)}}{2} - \frac{3}{(2+2\,I\,\sqrt{53})^{(1/3)}}\right)$$

Nótese que las soluciones no están dadas como números de punto flotante; anteriormente se habia mencionado que Maple generalmente maneja las expresiones numéricas usando aritmética racional, a menos que se le dé la indicación de usar números de punto flotante. Una forma de obtener las soluciones anteriores con punto flotante es usando la función **map**, su sintaxis es:

map(función, lista de datos);

map recibe como primer argumento una función o instrucción de Maple y la aplica sobre cada uno de los elementos de la lista que aparece como segundo argumento. Podemos utilizar esto para aplicar **evalf** a cada una de las soluciones anteriores:

```
> map(evalf, [%]);
```

$$[1{,}174833928 - 0{,}2\,10^{-9}\,I,\ -3{,}063415448 + 0.\,I,\ -1{,}111418480 + 0.\,I]$$

También podemos utilizar **solve** para resolver ecuaciones que contienen valores indeterminados:

```
> ec3 := a*x^2 + b*x + c = d;
```

$$ec3 := a\,x^2 + b\,x + c = d$$

```
> solve(ec3, x);
```

$$\frac{-b + \sqrt{b^2 - 4\,a\,c + 4\,a\,d}}{2\,a}, \frac{-b - \sqrt{b^2 - 4\,a\,c + 4\,a\,d}}{2\,a}$$

Nótese que obtenemos en este caso las soluciones generales para una ecuación de segundo grado.
Veamos otros ejemplos:

```
> ec4 := a*x^2*b^x*y + c*y^2 = d;
```

$$ec4 := a\,x^2\,b^x\,y + c\,y^2 = d$$

```
> solve(ec4);
```

$$\{d = a\,x^2\,b^x\,y + c\,y^2,\ a = a,\ x = x,\ y = y,\ b = b,\ c = c\}$$

```
> ec5 := 4*x + 5*y = 25;
```

$$ec5 := 4\,x + 5\,y = 25$$

```
> solve(ec5, x);
```

$$-\frac{5\,y}{4} + \frac{25}{4}$$

Una forma de verificar que tales soluciones son las correctas es sustituyendo cada una de ellas en la ecuación original. Por ejemplo, obtengamos las soluciones de la siguiente ecuación:

```
> ec6 := x^2 + 2*x + 3 = 7;
```

$$ec6 := x^2 + 2\,x + 3 = 7$$

```
> solve(ec6, x);
```

$$-1 + \sqrt{5},\ -1 - \sqrt{5}$$

Podemos crear una lista con estas soluciones de la siguiente forma:

```
> lsol := [%];
```

$$lsol := [\,-1 + \sqrt{5},\ -1 - \sqrt{5}\,]$$

Para acceder a cada uno de los elementos de la lista usamos las siguientes instrucciones:

```
> lsol[1]; # la primera solución contenida en la lista
```

$$-1 + \sqrt{5}$$

```
> lsol[2]; # la segunda solución
```

$$-1 - \sqrt{5}$$

Ahora, para comprobar que tales soluciones son correctas usamos la instrucción **subs** y los elementos de las lista de la siguiente forma:

```
> subs(x = lsol[1], ec6);
```

$$(\,-1 + \sqrt{5}\,)^2 + 1 + 2\,\sqrt{5} = 7$$

Es necesario utilizar la función expand para que lleve a cabo la exponenciación:

```
> expand(%);
```

$$7 = 7$$

Lo mismo con la segunda solución:

```
> expand(subs(x = lsol[2], ec6));
```

$$7 = 7$$

Como podemos ver, en este caso, Maple nos devolvio soluciones exactas, que al sustituirlas nos generan números exactos. En ocasiones, al sustituir tales soluciones se obtienen resultados complicados y es necesario manipular las expresiones de alguna forma (o de varias). Por ejemplo, resolvamos **ec2**:

```
> ec2;
```

$$2\,x^3 - 8\,x^2 + 3\,x + 8 = 4\,x^3 - 2\,x^2$$

```
> lsol2 := [solve(ec2, x)]; # creamos la lista de soluciones
```

$$lsol2 := [\frac{(2 + 2\,I\,\sqrt{53})^{(1/3)}}{2} + \frac{3}{(2 + 2\,I\,\sqrt{53})^{(1/3)}} - 1, -\frac{(2 + 2\,I\,\sqrt{53})^{(1/3)}}{4} - \frac{3}{2\,(2 + 2\,I\,\sqrt{53})^{(1/3)}} - 1 + \frac{1}{2}\,I\,\sqrt{3}\left(\frac{(2 + 2\,I\,\sqrt{53})^{(1/3)}}{2} - \frac{3}{(2 + 2\,I\,\sqrt{53})^{(1/3)}}\right), -\frac{(2 + 2\,I\,\sqrt{53})^{(1/3)}}{4} - \frac{3}{2\,(2 + 2\,I\,\sqrt{53})^{(1/3)}} - 1 - \frac{1}{2}\,I\,\sqrt{3}\left(\frac{(2 + 2\,I\,\sqrt{53})^{(1/3)}}{2} - \frac{3}{(2 + 2\,I\,\sqrt{53})^{(1/3)}}\right)]$$

Ahora, procederemos a sustituir tales soluciones:

```
> subs(x = lsol2[1], ec2);
```

$$2\ \%1^3 - 8\ \%1^2 + \frac{3\,(2+2\,I\,\sqrt{53})^{(1/3)}}{2} + \frac{9}{(2+2\,I\,\sqrt{53})^{(1/3)}} + 5 = 4\ \%1^3 - 2\ \%1^2$$

$$\%1 := \frac{(2+2\,I\,\sqrt{53})^{(1/3)}}{2} + \frac{3}{(2+2\,I\,\sqrt{53})^{(1/3)}} - 1$$

Aplicamos **expand**:

```
> expand(%);
```

$$-\frac{93}{2} + \frac{1}{2}\,I\,\sqrt{53} + 17\,(2+2\,I\,\sqrt{53})^{(1/3)} - \frac{7\,(2+2\,I\,\sqrt{53})^{(2/3)}}{2} + \frac{102}{(2+2\,I\,\sqrt{53})^{(1/3)}} + \frac{54}{2+2\,I\,\sqrt{53}}$$
$$- \frac{126}{(2+2\,I\,\sqrt{53})^{(2/3)}} = -47 + \sqrt{53}\,I + 17\,(2+2\,I\,\sqrt{53})^{(1/3)} - \frac{7\,(2+2\,I\,\sqrt{53})^{(2/3)}}{2}$$
$$+ \frac{102}{(2+2\,I\,\sqrt{53})^{(1/3)}} + \frac{108}{2+2\,I\,\sqrt{53}} - \frac{126}{(2+2\,I\,\sqrt{53})^{(2/3)}}$$

Y a continución aplicamos la instrucción **simplify** para que simplifique esta expresión:

```
> simplify(%);
```

$$-\frac{-34\,I\,\sqrt{53} + 46\,(2+2\,I\,\sqrt{53})^{(2/3)} - 95\,(2+2\,I\,\sqrt{53})^{(1/3)} + 7\,I\,(2+2\,I\,\sqrt{53})^{(1/3)}\,\sqrt{53} + 92}{(2+2\,I\,\sqrt{53})^{(2/3)}} =$$
$$-\frac{-34\,I\,\sqrt{53} + 46\,(2+2\,I\,\sqrt{53})^{(2/3)} - 95\,(2+2\,I\,\sqrt{53})^{(1/3)} + 7\,I\,(2+2\,I\,\sqrt{53})^{(1/3)}\,\sqrt{53} + 92}{(2+2\,I\,\sqrt{53})^{(2/3)}}$$

Y finalmente aplicamos **evalf**, a través de **map**, para obtener un resultado:

```
> map(evalf, [%]);
```

$$[3{,}725716946 - 0{,}7316337726\,10^{-8}\,I = 3{,}725716946 - 0{,}7316337726\,10^{-8}\,I]$$

Una vez hecho esto, podemos ver que ambos elementos de la relación son iguales. Otra forma de hacer esta comprobación es utilizar **rhs** y **lhs** para extraer los elementos derecho e izquierdo respectivamente y restarlos:

```
> lhs(%%) - rhs(%%);
```

$$0$$

De esta manera, comprobamos que la primera solución es correcta. Con la segunda solución procederemos de la siguiente forma:

```
> subs(x = lsol2[2], ec2);
```

$$2\ \%2^3 - 8\ \%2^2 - \frac{3\,(2+2\,I\,\sqrt{53})^{(1/3)}}{4} - \frac{9}{2\,(2+2\,I\,\sqrt{53})^{(1/3)}} + 5 + \frac{3}{2}\,I\,\sqrt{3}\ \%1 =$$
$$4\ \%2^3 - 2\ \%2^2$$

$$\%1 := \frac{(2+2\,I\,\sqrt{53})^{(1/3)}}{2} - \frac{3}{(2+2\,I\,\sqrt{53})^{(1/3)}}$$

$$\%2 := -\frac{(2+2\,I\,\sqrt{53})^{(1/3)}}{4} - \frac{3}{2\,(2+2\,I\,\sqrt{53})^{(1/3)}} - 1 + \frac{1}{2}\,I\,\sqrt{3}\ \%1$$

```
> evalc(%); # evaluación compleja
```

$$2\ \%3^3 - 8\ \%3^2 - \frac{3}{2}\sqrt{6}\cos(\ \%1) + 5 - \frac{3}{2}\ \%2 = 4\ \%3^3 - 2\ \%3^2$$
$$\%1 := \frac{1}{3}\arctan(\sqrt{53})$$
$$\%2 := \sqrt{3}\sqrt{6}\sin(\ \%1)$$
$$\%3 := -\frac{1}{2}\sqrt{6}\cos(\ \%1) - 1 - \frac{1}{2}\ \%2$$

```
> lhs(%) - rhs(%);
```

$$-2\left(-\frac{1}{2}\sqrt{6}\cos(\ \%1) - 1 - \frac{1}{2}\sqrt{3}\sqrt{6}\sin(\ \%1)\right)^3 - 6\left(-\frac{1}{2}\sqrt{6}\cos(\ \%1) - 1 - \frac{1}{2}\sqrt{3}\sqrt{6}\sin(\ \%1)\right)^2$$
$$-\frac{3}{2}\sqrt{6}\cos(\ \%1) + 5 - \frac{3}{2}\sqrt{3}\sqrt{6}\sin(\ \%1)$$
$$\%1 := \frac{1}{3}\arctan(\sqrt{53})$$

```
> simplify(%);
```

$$0$$

Ahora, hagamos lo mismo con la tercera solución:

```
> evalc(subs(x = lsol2[3], ec2));
```

$$2\ \%3^3 - 8\ \%3^2 - \frac{3}{2}\sqrt{6}\cos(\ \%1) + 5 + \frac{3}{2}\ \%2 = 4\ \%3^3 - 2\ \%3^2$$
$$\%1 := \frac{1}{3}\arctan(\sqrt{53})$$
$$\%2 := \sqrt{3}\sqrt{6}\sin(\ \%1)$$
$$\%3 := -\frac{1}{2}\sqrt{6}\cos(\ \%1) - 1 + \frac{1}{2}\ \%2$$

```
> simplify(lhs(%) - rhs(%));
```

$$0$$

De esta forma comprobamos que tales soluciones son correctas. Otras instrucciones utilizadas para resolver ecuaciones son:

- **isolve**. Resuelve ecuaciones, desplegando solo las soluciones enteras. Por ejemplo :

  ```
  > isolve(x^3 + x^2 - x + 8 = 9);
  ```

  $$\{x = -1\},\ \{x = 1\}$$

  En caso de que la ecuación no tenga soluciones enteras, **isolve** no muestra ningun resultado.

- **rsolve**. Resuelve ecuaciones de recurrencia.
- **msolve**. Resuelve ecuaciones en los enteros modulo m.
- **linsolve**. Resuelve ecuaciones matriciales. Esta función pertenece al paquete **linalg**, para usarla debe cargarse con **with** o bien usar la forma: **linalg[linsolve]**.
- **dsolve**. Resuelve ecuaciones diferenciales.

Consúltense las páginas de ayuda de ***rsolve*** y ***msolve*** para mayores referencias.

2.2. Soluciones aproximadas

Maple nos proporciona una forma de obtener soluciones de ecuaciones en forma numérica (aproximaciones de punto flotante). Esto se puede hacer a través de la instrucción **fsolve**. Por ejemplo, definamos la siguiente ecuación:

```
> eq1 := 2*x^2 + 4*x = 11;
```

$$eq1 := 2\,x^2 + 4\,x = 11$$

Utilicemos **fsolve** para obtener las soluciones:

```
> fsolve(eq1, x);
```

$$-3{,}549509757,\ 1{,}549509757$$

Veamos que sucede al sustituir estas soluciones en la ecuación:

```
> lista := [fsolve(eq1, x)];
```

$$lista := [-3{,}549509757,\ 1{,}549509757]$$

```
> subs(x = lista[1], eq1);
```

$$11{,}00000001 = 11$$

Ahora intentemos con la segunda solución:

```
> subs(x = lista[2], eq1);
```

$$11{,}00000000 = 11$$

Podemos ver que la primera solución tiene un pequeño margen de error, pues en realidad se trata de una aproximación numérica. En general, esta función no puede obtener soluciones exactas de ecuaciones, pero tiene la capacidad de calcular aproximaciones numéricas a estas soluciones. Obviamente no puede ser usada para calcular soluciones de ecuaciones en las cuales aparecen valores indeterminados.

Una de las desventajas de esta función es que no siempre encuentra soluciones complejas. Veamos un ejemplo:

```
> eq2 := 2*x^2 + 4*x = x - 9;
```

$$eq2 := 2\,x^2 + 4\,x = x - 9$$

Resolvamos primero utilizando **solve**:

```
> solve(eq2, x);
```

$$-\frac{3}{4} + \frac{3}{4}\,I\,\sqrt{7},\ -\frac{3}{4} - \frac{3}{4}\,I\,\sqrt{7}$$

Nótese que ambas soluciones son complejas, intentemos usando **fsolve**:

```
> fsolve(eq2, x);
```

En este caso la función no nos devuelve ninguna solución. Este problema se puede resolver utilizando la opción **complex** en **fsolve:**

```
> fsolve(eq2, x, complex);
```

$$-0{,}7500000000 - 1{,}984313483\,I,\ -0{,}7500000000 + 1{,}984313483\,I$$

Existe una opción conocida como **maxsols**, la cual nos permite indicar a **fsolve** cuantas soluciones queremos que nos muestre. Esta opción puede ser utilizada, por ejemplo, al resolver polinomios. Veamos un ejemplo:

```
> fsolve(23*x^4 + 10*x^3 - 28*x^2 + 17*x = 3, x, maxsols=2);
```

$$-1{,}556547150,\ 0{,}2933145038$$

Otra de las opciones que nos proporciona esta función es la poder incluir un intervalo; esto nos permite solicitar a **fsolve** unicamente aquellas soluciones que se encuentren dentro de dicho intervalo. Tal intervalo puede ser incluido en la forma **"a..b"**, **"x=a..b"** o bien **"{x=a..b, y=c..d, ...}"**. Estos intervalos se consideran cerrados y pueden incluir **infinity** o **-infinity**.

Por ejemplo, para obtener todas las soluciones positivas de la expresión anterior :

```
> fsolve(23*x^4 + 10*x^3 - 28*x^2 + 17*x = 3, x=0..infinity);
```

$$0{,}2933145038$$

otra forma de ejecutar esta instrucción es:

```
> fsolve(23*x^4 + 10*x^3 - 28*x^2 + 17*x = 3, x, 0..infinity);
```

$$0{,}2933145038$$

Otra opción de **fsolve** que puede ser útil en estos casos es **avoid**, con la cual le indicamos a esta función que calcule las soluciones pero sin tomar en cuenta el intervalo especificado por la opción **avoid**.

3. Interpretación de la expresión RootOf

Existen casos en los cuales Maple despliega expresiones en términos de una función conocida como **RootOf**. Tal función es utilizada para representar todas las raices de una ecuación que depende de una variable. Aunque tiene algunos otros usos, Maple utiliza esta función para expresar las soluciones de ecuaciones y sistemas de ecuaciones polinomiales. Por ejemplo:

```
> ecn := x - y = sin(x);
```

$$ecn := x - y = \sin(x)$$

```
> solve(ecn, x);
```

$$\mathrm{RootOf}(\_Z - y - \sin(\_Z))$$

Maple identifica de manera simbólica a la raiz de la ecuación con la expresión **RootOf()**. Veamos que sucede en el siguiente caso:

```
> pol := a*x^5 + b*x^2 - x + 1 = 0;
```

$$pol := a\,x^5 + b\,x^2 - x + 1 = 0$$

```
> solve(pol, x);
```

$$\mathrm{RootOf}(a\,\_Z^5 + b\,\_Z^2 - \_Z + 1)$$

Ahora, sustituimos el resultado obtenido en la ecuación:

```
> subs(x =%, lhs(pol) = rhs(pol));
```

$$a\,\mathrm{RootOf}(a\,\_Z^5 + b\,\_Z^2 - \_Z + 1)^5 + b\,\mathrm{RootOf}(a\,\_Z^5 + b\,\_Z^2 - \_Z + 1)^2$$
$$- \mathrm{RootOf}(a\,\_Z^5 + b\,\_Z^2 - \_Z + 1) + 1 = 0$$

```
> simplify(%);
```

$$0 = 0$$

Como podemos ver, la expresión **RootOf()** efectivamente es manejada como una raiz de la ecuación. Es común que esta expresión aparezca en ecuaciones algebraicas que no pueden ser resueltas en forma exacta (o cuya solución es demasiado complicada). Utilizando la instrucción **allvalues**, en algunos casos se pueden obtener todos los valores posibles de una expresión en la cual se tiene involucrada **RootOf**. Por ejemplo:

```
> solve(3*x^4+5*x=2,x);
```

$$\mathrm{RootOf}(\%1,\ index=1),\ \mathrm{RootOf}(\%1,\ index=2),\ \mathrm{RootOf}(\%1,\ index=3),$$
$$\mathrm{RootOf}(\%1,\ index=4)$$
$$\%1 := 3\ \_Z^4 + 5\ \_Z - 2$$

```
> map(allvalues, [%]);
```

$$\left[-\frac{\sqrt{6}\sqrt{\%2}}{12}+\frac{\sqrt{6}\sqrt{\frac{-\sqrt{\%2}\ \%1+32\sqrt{\%2}+20\sqrt{6}\ \%3}{(300+4\sqrt{7673})^{(1/3)}\sqrt{\%2}}}}{12},\right.$$
$$\frac{\sqrt{6}\sqrt{\%2}}{12}+\frac{1}{12}I\sqrt{\frac{6\sqrt{\%2}\ \%1-192\sqrt{\%2}+120\sqrt{6}\ \%3}{(300+4\sqrt{7673})^{(1/3)}\sqrt{\%2}}},$$
$$-\frac{\sqrt{6}\sqrt{\%2}}{12}-\frac{\sqrt{6}\sqrt{\frac{-\sqrt{\%2}\ \%1+32\sqrt{\%2}+20\sqrt{6}\ \%3}{(300+4\sqrt{7673})^{(1/3)}\sqrt{\%2}}}}{12},$$
$$\left.\frac{\sqrt{6}\sqrt{\%2}}{12}-\frac{1}{12}I\sqrt{\frac{6\sqrt{\%2}\ \%1-192\sqrt{\%2}+120\sqrt{6}\ \%3}{(300+4\sqrt{7673})^{(1/3)}\sqrt{\%2}}}\right]$$
$$\%1 := (300+4\sqrt{7673})^{(2/3)}$$
$$\%2 := \frac{\%1-32}{(300+4\sqrt{7673})^{(1/3)}}$$
$$\%3 := (300+4\sqrt{7673})^{(1/3)}$$

La función **allvalues** procede de la siguiente forma al obtener los valores de **RootOf**:

- Todas las raices que puedan ser obtenidas de manera exacta, son calculadas utilizando la función **solve**. Por ejemplo, las raices de polinomios de grado menor o igual a cuatro pueden obtenerse de esta forma.
- Para aquellas raices que no puedan determinarse de manera exacta, **allvalues** utiliza **fsolve** para calcularlas en forma numérica. Obviamente, en este caso no se admiten constantes indeterminadas en la expresión.

4. Interpretación gráfica de la solución de una ecuación

Gráficamente, las soluciones de una ecuación de la siguiente forma:

```
> ec := 4*x + 5*x^2 - 2 = 0;
```

$$ec := 4\,x + 5\,x^2 - 2 = 0$$

corresponden a los valores de **x** en los cuales la ecuación se hace cero (es decir, corresponden a ***"los ceros"*** de la ecuación). Primero, obtendremos dichas soluciones:

```
> solve(ec, x);
```

$$-\frac{2}{5}+\frac{\sqrt{14}}{5},\ -\frac{2}{5}-\frac{\sqrt{14}}{5}$$

```
> evalf(%);
```

$$0{,}3483314774,\ -1{,}148331477$$

Ahora, graficamos la ecuación:

```
> plot(4*x + 5*x^2 - 2, x=-1.5..0.5, -3..1);
```



Podemos ver que en la gráfica aparecen aproximadas las soluciones. De la misma forma, para la expresión:

```
> ec2 := x^2 + 9 = x - x^3;
```

$$ec2 := x^2 + 9 = x - x^3$$

```
> solve(ec2, x);
```

$$-\frac{(127+3\sqrt{1785})^{(1/3)}}{3} - \frac{4}{3\,(127+3\sqrt{1785})^{(1/3)}} - \frac{1}{3},\ \frac{(127+3\sqrt{1785})^{(1/3)}}{6} + \frac{2}{3\,(127+3\sqrt{1785})^{(1/3)}} - \frac{1}{3} + \frac{1}{2}\,I\sqrt{3}\left(-\frac{(127+3\sqrt{1785})^{(1/3)}}{3} + \frac{4}{3\,(127+3\sqrt{1785})^{(1/3)}}\right),\ \frac{(127+3\sqrt{1785})^{(1/3)}}{6} + \frac{2}{3\,(127+3\sqrt{1785})^{(1/3)}} - \frac{1}{3} - \frac{1}{2}\,I\sqrt{3}\left(-\frac{(127+3\sqrt{1785})^{(1/3)}}{3} + \frac{4}{3\,(127+3\sqrt{1785})^{(1/3)}}\right)$$

```
> evalf(%);
```

$$-2{,}654249157,\ 0{,}8271245787 - 1{,}645191285\,I,\ 0{,}8271245787 + 1{,}645191285\,I$$

Desplegamos la gráfica:

```
> plot(x^2 + 9 - x + x^3, x=-3..3, y=-3..12);
```



En este caso, podemos ver que solo se tiene una solución real, pues la gráfica solo atraviesa una vez el eje (solo existe un cero), mientras que existen también dos raices complejas.

5. Sistemas de ecuaciones

Para resolver sistemas de ecuaciones, utilizamos las mismas funciones **solve** y **fsolve**, pero aplicadas a conjuntos de ecuaciones; con esto obtenemos una solución que satisface todas las ecuaciones del sistema de manera simultanea. Para definir un sistema de ecuaciones podemos proceder de la siguiente forma:

```
> ec1 := x + y + z = 9;
```

$$ec1 := x + y + z = 9$$

```
> ec2 := 4*y + 3*z = 6;
```

$$ec2 := 4\,y + 3\,z = 6$$

```
> ec3 := x - y + 2*z = 4;
```

$$ec3 := x - y + 2\,z = 4$$

Y a continuación colocamos todas las ecuaciones en forma de un conjunto para obtener su solución, por ejemplo a través de **solve**:

solve({ec1, ec2, ec3, ...}, {x1, x2, x3, ...});

Donde **"ec1, ec2, ec3, ..."** son la ecuaciones que pertenecen al sistema y **"x1, x2, x3, ..."** son las incógnitas con respecto a las que se desea resolver. De hecho, éstas últimas pueden omitirse y entonces Maple tratará de obtener una solución para cada una de las incógnitas presentes. Así, podemos resolver el sistema formado por **"ec1"**, **"ec2"** y **"ec3"** de la siguiente forma:

```
> solve({ec1, ec2, ec3}, {x, y, z});
```

$$\{y = \frac{21}{10},\, x = \frac{77}{10},\, z = \frac{-4}{5}\}$$

La interpretación gráfica de esta solución es el punto en el cual se intersectan los planos definidos por cada una de las ecuaciones. Podemos utilizar la instrucción **implicitplot3d** del paquete **plots** para desplegar su gráfica:

```
> plots[implicitplot3d]({x + y + z - 9, 2*z + 4*y + z - 6,
> x - y + 2*z - 4}, x=-10..10, y=-10..10, z=-10..10);
```



Para aquellos sistemas en los cuales Maple no puede determinar soluciones exactas, también es posible usar **fsolve** para calcular una aproximación numérica a las soluciones. Por ejemplo:

```
> eq1 := x + y + z = cos(x);
```

$$\mathit{eq1} := x + y + z = \cos(x)$$

```
> eq2 := 5*x + 4*y + z = 3*x + y;
```

$$\mathit{eq2} := 5\,x + 4\,y + z = 3\,x + y$$

```
> eq3 := 3*x - y - z = 9;
```

$$\mathit{eq3} := 3\,x - y - z = 9$$

```
> solve({eq1, eq2, eq3}, {x, y, z});
```

$$\{y = -\frac{5}{2}\,\mathrm{RootOf}(4\,\_Z - 9 - \cos(\_Z)) + \frac{9}{2},\ x = \mathrm{RootOf}(4\,\_Z - 9 - \cos(\_Z)),$$
$$z = \frac{11}{2}\,\mathrm{RootOf}(4\,\_Z - 9 - \cos(\_Z)) - \frac{27}{2}\}$$

En este caso, solve no puede obtener una solución exacta, pero **fsolve** sí puede calcular una aproximación:

```
> fsolve({eq1, eq2, eq3}, {x, y, z});
```

$$\{x = 2{,}119586203,\ y = -0{,}7989655084,\ z = -1{,}842275882\}$$

Nuevamente recuérdese que tales soluciones no son exactas, en realidad se trata de aproximaciones que pueden presentar un cierto grado de error.

6. Solución de sistemas de ecuaciones lineales con matrices

En el caso de los sistemas de ecuaciones lineales, estas pueden ser resueltas utilizando operaciones matriciales. El tema de matrices se tratara posteriormente, pero haremos una pequeña introducción para poder resolver sistemas de ecuaciones por este método.

6.1. Matrices

6.1.1. Definicion de matrices

Una forma de manejar matrices en Maple es por medio del paquete **LinearAlgebra**, el cual contiene un conjunto de funciones para manejo de vectores y matrices, así como para realizar diversos cálculos de Álgebra Lineal. Este paquete puede cargarse en memoria por medio de la instrucción **with**.

Por otro lado, una manera de definir una matriz es por medio de la instruccion **Matrix** (la cual no pertenece a ningun paquete). Ésta nos proporciona varias formas de crear matrices, una de ellas es:

Matrix(n, m, [[a11, a12, a13, ...a1m], [a21, ...,a2m], ...,[an1, ..., anm]]);

Donde **n** y **m** indican el número de renglones y columnas de la matriz, respectivamente, mientras que **"[a11, a12, a13,...a1m], [a21,...,a2m], ...,[an1,..., anm]"** son los elementos de cada uno de los renglones (cada uno de éstos debe colocarse como una lista). En caso de que los elementos no sean especificados, la matriz es creada con todos sus elementos iguales a cero. Por ejemplo, la instrucción:

```
> A := Matrix(2, 2, [[1, 2], [3, 4]]);
```

$$A := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

define una matriz de dos renglones y dos columnas, cuyos elementos son 1, 2, 3, 4. Veamos otro ejemplo:

```
> B := Matrix(1, 2, [3, 4]);
```

$$B := \begin{bmatrix} 3 & 4 \end{bmatrix}$$

Los elementos de una matriz pueden estar dados no solo por números, también pueden incluirse expresiones algebraicas, aritméticas, constantes y variables simbólicas. Por ejemplo:

```
> C := Matrix(4, 4,
> [[1, 2, 3, 4], [sin(x), 4, 9, Pi^2], [8, sqrt(2), 3, 1], [1, 2, 3,
> 4]]);
```

$$C := \begin{bmatrix} 1 & 2 & 3 & 4 \\ \sin(x) & 4 & 9 & \pi^2 \\ 8 & \sqrt{2} & 3 & 1 \\ 1 & 2 & 3 & 4 \end{bmatrix}$$

Para acceder a un elemento de una matriz, utilizamos la siguiente expresión:

M[a, b]

Donde **M** es el nombre de la matriz, **a** es el renglón donde se encuentra el elemento al cual se desea acceder y **b** es la columna del mismo. Por ejemplo para acceder al elemento del segundo renglón y la segunda columna de la matriz **C** utilizamos:

```
> C[2,2];
```

$$4$$

De la misma forma podemos, por ejemplo, modificar los elementos de la matriz **C**:

```
> C[1,1]:=9;
```

$$C_{1,1} := 9$$

```
> C[1,3]:=.8764532;
```

$$C_{1,3} := 0{,}8764532$$

6.1.2. Evaluación, suma y multiplicación de matrices

6.1.3. Evaluacion

En el caso de las matrices creadas con la función **Matrix**, para desplegar sus elementos es suficiente teclear su nombre. Definamos la matriz **"E"**:

```
> E := Matrix(4, 4, [[3, 4, 6, 4], [8, cos(x), .867, Pi^2],
> [5, exp(2), 3, 1], [1, 2, 3, 4]]);
```

$$E := \begin{bmatrix} 3 & 4 & 6 & 4 \\ 8 & \cos(x) & 0{,}867 & \pi^2 \\ 5 & e^2 & 3 & 1 \\ 1 & 2 & 3 & 4 \end{bmatrix}$$

```
> E;
```

$$\begin{bmatrix} 3 & 4 & 6 & 4 \\ 8 & \cos(x) & 0{,}867 & \pi^2 \\ 5 & e^2 & 3 & 1 \\ 1 & 2 & 3 & 4 \end{bmatrix}$$

Como podemos ver, al colocar su nombre, Maple nos devuelve automáticamente los elementos.

6.1.4. Suma

La suma de matrices es la matriz formada con la suma de los elementos. Recuérdese que para llevar a cabo esta operación las matrices deben ser del mismo orden. Por ejemplo:

```
> M := Matrix(2, 2, [[3, 4], [78, 2]]);
```

$$M := \begin{bmatrix} 3 & 4 \\ 78 & 2 \end{bmatrix}$$

```
> N := Matrix(2, 2, [[43, 9], [1, 8]]);
```

$$N := \begin{bmatrix} 43 & 9 \\ 1 & 8 \end{bmatrix}$$

```
> P := Matrix(1, 2, [8, 2]);
```

$$P := \begin{bmatrix} 8 & 2 \end{bmatrix}$$

Para obtener la suma podemos utilizar el operador aritmético '+' de la siguiente forma:

```
> M + N;
```

$$\begin{bmatrix} 46 & 13 \\ 79 & 10 \end{bmatrix}$$

Cuando se llevan a cabo operaciones con matrices deben tenerse en consideración las dimensiones éstas. Por ejemplo, a continuación intentemos sumar **M + P** que tienen orden diferente (2x2 y 2x1):

```
> M + P;
```

```
Error, (in rtable/Sum) invalid arguments
```

Maple automáticamente verifica las dimensiones antes de ejecutar la operación.

Para calcular este tipo de sumas también puede usarse la función **evalm**, en la forma: **evalm(M + N)**; sin embargo es más sencillo usar directamente el operador aritmético.

6.1.5. Multiplicacion.

La multiplicacion de matrices (por ejemplo **A X**) no se efectua con * (reservado para multiplicacion de numeros y expresiones), tampoco puede calcularse por medio de **evalm**:

```
> M := Matrix(2, 2, [[3, 4], [78, 2]]);
```

$$M := \begin{bmatrix} 3 & 4 \\ 78 & 2 \end{bmatrix}$$

```
> N := Matrix(2, 2, [[43, 9], [1, 8]]);
```

$$N := \begin{bmatrix} 43 & 9 \\ 1 & 8 \end{bmatrix}$$

```
> P := Matrix(1, 2, [8, 2]);
```

$$P := \begin{bmatrix} 8 & 2 \end{bmatrix}$$

```
> evalm(M*P);
Error, (in rtable/Product) invalid arguments
```

Existen dos formas de multiplicar matrices, utilizando el operador **"."**, o también utilizando la instrucción **MatrixMatrixMultiply** del paquete **LinearAlgebra**:

```
> M.N;
```

$$\begin{bmatrix} 133 & 59 \\ 3356 & 718 \end{bmatrix}$$

```
> LinearAlgebra[MatrixMatrixMultiply](M, N);
```

$$\begin{bmatrix} 133 & 59 \\ 3356 & 718 \end{bmatrix}$$

Recuérdese que la multiplicacion de matrices no es conmutativa, es decir, en general se cumple que **M x N** es diferente de **N x M**. Ejemplo:

```
> M.N;
```

$$\begin{bmatrix} 133 & 59 \\ 3356 & 718 \end{bmatrix}$$

```
> N.M;
```

$$\begin{bmatrix} 831 & 190 \\ 627 & 20 \end{bmatrix}$$

6.1.6. Matrices Inversa e Identidad

Maple tiene la capacidad de calcular matrices inversas y de manejar matrices identidad. Dada una matriz ***"Q"***, podemos calcular su inversa por medio de la instrucción **MatrixInverse** (perteneciente al paquete **LinearAlgebra**), de la siguiente forma:

```
> Q := Matrix(2, 2, [[9, 2], [5, 7]]);
```

$$Q := \begin{bmatrix} 9 & 2 \\ 5 & 7 \end{bmatrix}$$

```
> Qinv := LinearAlgebra[MatrixInverse](Q);
```

$$Qinv := \begin{bmatrix} \frac{7}{53} & \frac{-2}{53} \\ \frac{-5}{53} & \frac{9}{53} \end{bmatrix}$$

Esta matriz inversa se representa como: $Q^{(-1)}$ (es decir, la matriz inversa de **Q**), tal que:

$$Q^{(-1)}\, Q = Id$$

donde ***Id*** es la ***"matriz identidad"***, la cual es de la misma dimensión que **Q** en este caso, y contiene unicamente el valor **"1"** en su diagonal y ceros en las otras posiciones. Además, esta matriz tiene la propiedad:

$$Q\, Id = Q$$
$$Id\, Q = Q$$

Esto para cualquier matriz **Q** (recuérdese que son importantes las dimensiones).

Verifiquemos si **Qinv** realmente es la inversa de **Q**:

```
> Qinv.Q;
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

```
> Q.Qinv;
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Una forma de generar una matriz identidad es por medio de la función **IdentityMatrix de LinearAlgebra**. Por ejemplo, crearemos una matriz identidad de 2 x 2:

```
> Id := LinearAlgebra[IdentityMatrix](2);
```

$$Id := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Multipliquemos **Id** por **Q**:

```
> Q;
```

$$\begin{bmatrix} 9 & 2 \\ 5 & 7 \end{bmatrix}$$

```
> Id.Q;
```

$$\begin{bmatrix} 9 & 2 \\ 5 & 7 \end{bmatrix}$$

```
> Q.Id;
```

$$\begin{bmatrix} 9 & 2 \\ 5 & 7 \end{bmatrix}$$

6.2. Solución de un sistema de dos ecuaciones lineales con dos incognitas

Dado un sistema de ecuaciones de la forma:

$$a_1\, x + b_1\, y = c_1$$
$$a_2\, x + b_2\, y = c_2$$

donde x, y son las incognitas, a_1, b_1, a_2, b_2 y c_1, c_2 son los términos independientes. Cada una de las dos ecuaciones puede interpretarse como una expresión o función cuya gráfica es una linea recta (despejando y en terminos de x). Por esta razón originalmente se conocen con el nombre de ***"ecuaciones lineales"***.

La solucion del sistema formado por estas ecuaciones se puede obtener aplicando **solve** (o **fsolve**) al conjunto de ecuaciones, en la forma:

solve({ec1, ec2}, {x, y});

La interpretacián gráfica de la solución es el punto de intersección de las rectas:

$y := -\frac{a_1\,x}{b_1} + \frac{c_1}{b_1}$, $y := -\frac{a_2\,x}{b_2} + \frac{c_2}{b_2}$

Otra forma de resolver este sistema (sin utilizar **solve** o **fsolve**) es por medio de matrices.

El sistema general de dos ecuaciones lineales inhomogeneas, planteado anteriormente, es equivalente a la ecuación matricial:

$$\mathbf{A}\;\mathbf{X} = \mathbf{Y}$$

Donde **A** es la matriz formada por los coeficientes de las ecuaciones, **X** es la matrix formada por las incognitas y **Y** es la matriz formada por los términos independientes. Veamos un ejemplo particular, para el siguiente sistema de ecuaciones lineales:

$3\,x + 2\,y = 1$

$3\,x - 2\,y = 2$

Su ecuación matricial está dada por: $A\,X = Y$, donde:

$$A = \begin{bmatrix} 3 & 2 \\ 3 & -2 \end{bmatrix}$$

$$X = \begin{bmatrix} x \\ y \end{bmatrix}$$

$$Y = \begin{bmatrix} 1 \\ 2 \end{bmatrix}$$

Para resolver esta ecuación matricial necesitamos encontrar la inversa de **A**, una vez hecho esto multiplicamos cada lado de la ecuación matricial por esta inversa y obtenemos:

$$A^{(-1)}\,A\,X = A^{(-1)}\,Y$$
$$Id\,X = A^{(-1)}\,Y$$

De donde:

$$X = A^{(-1)}\,Y$$

Con lo cual obtenemos la solución. Antes de proceder nos aseguraremos de cargar el paquete **LinearAlgebra**:

```
> with(LinearAlgebra):
```

Ahora definimos **A**, **X**, **Y** y calculamos la inversa de **A**:

```
> A := Matrix(2, 2, [[3, 2], [3, -2]]);
```

$$A := \begin{bmatrix} 3 & 2 \\ 3 & -2 \end{bmatrix}$$

```
> X := Matrix(2, 1, [[x],[y]]);
```

$$X := \begin{bmatrix} x \\ y \end{bmatrix}$$

```
> Y := Matrix(2, 1, [[1], [2]]);
```

$$Y := \begin{bmatrix} 1 \\ 2 \end{bmatrix}$$

```
> Inv_A := MatrixInverse(A);
```

$$Inv\_A := \begin{bmatrix} \frac{1}{6} & \frac{1}{6} \\ \frac{1}{4} & \frac{-1}{4} \end{bmatrix}$$

Despues definimos la ecuación matricial:

```
> ecmat := A.X = Y;
```

$$ecmat := \begin{bmatrix} 3\,x + 2\,y \\ 3\,x - 2\,y \end{bmatrix} = \begin{bmatrix} 1 \\ 2 \end{bmatrix}$$

Y finalmente obtenemos la solución multiplicando ambos lados por la matriz inversa de **A**:

```
> sol := Inv_A.lhs(ecmat) = Inv_A.rhs(ecmat);
```

$$sol := \begin{bmatrix} x \\ y \end{bmatrix} = \begin{bmatrix} \frac{1}{2} \\ \frac{-1}{4} \end{bmatrix}$$

Por lo tanto la solución es: $x = \frac{1}{2}$, $y = -\frac{1}{4}$. Comparemos con el resultado generado por **solve:**

```
> solve({3*x + 2*y = 1, 3*x - 2*y = 2}, {x, y});
```

$$\{y = \frac{-1}{4},\, x = \frac{1}{2}\}$$

Esta solución fue calculada llevando a cabo cada uno de las operaciones matriciales; sin embargo, también puede ser obtenida por medio de la instrucción **LinearSolve** del paquete **Linear Algebra**, la cual resuelve ecuaciones matriciales de la forma **A.x** = **Y**. La sintaxis para realizar esta operación es:

LinearSolve(A, Y);

como se muestra a continuación:

```
> LinearSolve(A, Y);
```

$$\begin{bmatrix} \frac{1}{2} \\ \frac{-1}{4} \end{bmatrix}$$

En el caso de estas ecuaciones, la solución se puede interpretar graficamente como el punto en el cual se intersectan las rectas definidas por cada una de ellas. Para poder visualizar esto primero despejaremos ambas ecuaciones en términos de **y**:

```
> ec1 := solve(3*x + 2*y = 1, y);
```

$$ec1 := -\frac{3\,x}{2} + \frac{1}{2}$$

```
> ec2 := solve(3*x - 2*y = 2, y);
```

$$ec2 := \frac{3\,x}{2} - 1$$

Ahora deplegaremos las gráficas para poder apreciar el punto de intersección:

```
> plot({ec1, ec2}, x=-1..1);
```



6.3. Sistemas de tres ecuaciones lineales con tres incognitas

Las ecuaciones de la forma:

$$a_1\, x + b_1\, y + c_1\, z = d_1$$

son llamadas ecuaciones lineales. Nótese que las tres incognitas **x**, **y**, **z** aparecen elevadas a la potencia **"1"** (es decir, no aparecen términos como x^2, $\sqrt{y}$, $\sin(z)$**,** ni productos como $x\, y$, $z\, x$, etc. En el caso anterior, teniamos:

$$a_1\, x + b_1\, y = c_1$$

Despejando **y** obteniamos una expresión (o función) cuya gráfica era una linea recta, de ahi el nombre ***"lineal"***. En el caso de tres incognitas, si despejamos una de ellas (por ejemplo **z**), tendriamos:

$$z = -\frac{a_1\, x}{c_1} - \frac{b_1\, y}{c_1} - \frac{d_1}{c_1}$$

cuya grafica es un plano en el espacio. Aunque el plano es una generalizacion de una linea recta, a estas ecuaciones las seguimos llamando ***"lineales"*** y no ***"planares"***.

La forma de resolver sistemas de tres ecuaciones con tres incognitas es equivalente a la solución de sistemas de dos ecuaciones con dos incognitas.

Veamos un ejemplo con un sistema de tres ecuaciones lineales con tres incognitas:

```
> ec1 := 3*x + 2*y - z = 0;
```

$$ec1 := 3\, x + 2\, y - z = 0$$

```
> ec2 := -2*x - 3*y + z/2 = -1;
```

$$ec2 := -2\, x - 3\, y + \frac{z}{2} = -1$$

```
> ec3 := -x - y - z = 1;
```

$$ec3 := -x - y - z = 1$$

Podemos utilizar la instrucción **coeffs** para extraer los coeficientes de la parte izquierda de cada relación. Por ejemplo:

```
> coeffs(lhs(ec1));
```

$$3,\ 2,\ -1$$

A continuación definiremos los elementos de la ecuación matricial y resolveremos ésta por medio de la instrucción **LinearSolve**:

```
>  A := Matrix(3, 3,
>  [[coeffs(lhs(ec1))], [coeffs(lhs(ec2))], [coeffs(lhs(ec3))]]);
```

$$A := \begin{bmatrix} 3 & 2 & -1 \\ -2 & -3 & \frac{1}{2} \\ -1 & -1 & -1 \end{bmatrix}$$

```
>  Y := Matrix(3, 1, [[rhs(ec1)], [rhs(ec2)], [rhs(ec3)]]);
```

$$Y := \begin{bmatrix} 0 \\ -1 \\ 1 \end{bmatrix}$$

Ahora obtenemos la solución:

```
>  LinearSolve(A, Y);
```

$$\begin{bmatrix} \frac{-10}{13} \\ \frac{9}{13} \\ \frac{-12}{13} \end{bmatrix}$$

Comparemos con la solución obtenida por **solve**:

```
>  solve({ec1, ec2, ec3}, {x, y, z});
```

$$\{z = \frac{-12}{13},\, y = \frac{9}{13},\, x = \frac{-10}{13}\}$$

Para trazar la grafica de la expresion que define el plano obtenido al despejar **z**, usamos la instruccion para graficas tridimensionales **plot3d**, cuya sintaxis es:

plot3d(expresion, x=rango, y=rango);

Primero despejamos **z**, por ejemplo en la ecuación 1:

```
>  plano_1 := solve(ec1, z);
```

$$plano\_1 := 3\,x + 2\,y$$

Y a continuación graficamos:

```
> plot3d(plano_1, x=-5..5, y=-5..5);
```



Lo mismo para los otros planos:

```
> plano_2 := solve(ec2, z); plano_3 := solve(ec3, z);
```

$$plano\_2 := 4\,x + 6\,y - 2$$

$$plano\_3 := -x - y - 1$$

```
> plot3d({plano_1, plano_2, plano_3}, x=-5..5, y=-5..5,
> axes=BOXED);
```



En este caso, la solución se puede interpretar como el punto en el cual se intersectan los tres planos en el espacio.

7. Gráficas de sistemas de ecuaciones

7.1. Dos ecuaciones lineales con dos incognitas

Considérese el siguiente sistema de dos ecuaciones lineales:

```
> sist_lin := 3*x - y = 4, -7*x + 8*y = 1;
```

$$sist\_lin := 3\,x - y = 4,\ -7\,x + 8\,y = 1$$

Podemos calcular la solución a través de **solve**:

```
> solve({sist_lin}, {x, y});
```

$$\{x = \frac{33}{17},\ y = \frac{31}{17}\}$$

La interpretación gráfica de dicha solución es el punto común de las ecuaciones:

$$\begin{array}{c} 3\,x\ -\ y\ =\ 4 \\ -7\,x + 8\,y = 1 \end{array}$$

Esto puede visualizarse graficando ambas funciones y viendo su punto de interseccion en la gráfica. Podemos usar la función **implicitplot** del paquete **plots**.

```
> plots[implicitplot]({3*x - y = 4, -7*x + 8*y = 1}, x=-3..3,
> y=-3..3);
```



Este razonamiento se aplica tambien a ecuaciones no-lineales, por ejemplo:

```
> sist := 2*x^2 + 2*y - 1 = x, y^2 - 2*x + 1 = y;
```

$$sist := 2\,x^2 + 2\,y - 1 = x,\ y^2 - 2\,x + 1 = y$$

Al aplicar **solve**, resultan dos soluciones muy complicadas:

```
> solve({sist}, {x, y});
```

$$\{y = -1,\ x = \frac{3}{2}\},\ \{x = \frac{1}{2}\ \%1^2 + \frac{1}{2} - \frac{1}{2}\ \%1,\ y = \%1\}$$
$$\%1 := \mathrm{RootOf}(\_Z^3 - 3\ \_Z^2 + 5\ \_Z - 2,\ label = \_L10)$$

```
> allvalues(%[2]);
```

$$\left\{x = \frac{(-\frac{(108+12\sqrt{177})^{(1/3)}}{6} + \frac{4}{(108+12\sqrt{177})^{(1/3)}} + 1)^2}{2} + \frac{(108+12\sqrt{177})^{(1/3)}}{12}\right.$$

$$\left. - \frac{2}{(108+12\sqrt{177})^{(1/3)}},\ y = -\frac{(108+12\sqrt{177})^{(1/3)}}{6} + \frac{4}{(108+12\sqrt{177})^{(1/3)}} + 1\right\},\ \left\{x = \right.$$

$$\frac{(\frac{(108+12\sqrt{177})^{(1/3)}}{12} - \frac{2}{(108+12\sqrt{177})^{(1/3)}} + 1 + \frac{1}{2}I\sqrt{3}\ \%1)^2}{2}$$

$$-\frac{(108+12\sqrt{177})^{(1/3)}}{24} + \frac{1}{(108+12\sqrt{177})^{(1/3)}} - \frac{1}{4}I\sqrt{3}\ \%1,$$

$$\left. y = \frac{(108+12\sqrt{177})^{(1/3)}}{12} - \frac{2}{(108+12\sqrt{177})^{(1/3)}} + 1 + \frac{1}{2}I\sqrt{3}\ \%1\right\},\ \left\{x = \right.$$

$$\frac{(\frac{(108+12\sqrt{177})^{(1/3)}}{12} - \frac{2}{(108+12\sqrt{177})^{(1/3)}} + 1 - \frac{1}{2}I\sqrt{3}\ \%1)^2}{2}$$

$$-\frac{(108+12\sqrt{177})^{(1/3)}}{24} + \frac{1}{(108+12\sqrt{177})^{(1/3)}} + \frac{1}{4}I\sqrt{3}\ \%1,$$

$$\left. y = \frac{(108+12\sqrt{177})^{(1/3)}}{12} - \frac{2}{(108+12\sqrt{177})^{(1/3)}} + 1 - \frac{1}{2}I\sqrt{3}\ \%1\right\}$$

$$\%1 := -\frac{(108+12\sqrt{177})^{(1/3)}}{6} - \frac{4}{(108+12\sqrt{177})^{(1/3)}}$$

Una solucion verdaderamente aparatosa. Por otra parte, **fsolve** solo muestra una solución:

```
> fsolve({sist}, {x, y});
```

$$\{x = 1{,}500000000,\ y = -1{,}000000000\}$$

Veamos el aspecto de las dos soluciones graficamente:

```
> plots[implicitplot]({sist[1], sist[2]}, x=-3..3, y=-3..3);
```



7.2. Dos ecuaciones con tres incognitas

Veamos el siguiente ejemplo con dos ecuaciones lineales:

```
> sist_lin := 3*x - y + 2*z = 4, -7*x + 8*y - z/2 = 1;
```

$$sist\_lin := 3\,x - y + 2\,z = 4,\ -7\,x + 8\,y - \frac{z}{2} = 1$$

Aplicaremos **solve** y desplegaremos la gráfica del sistema.

```
> solve({sist_lin}, {x, y, z});
```

$$\{z = -\frac{34\,x}{31} + \frac{66}{31},\ x = x,\ y = \frac{25\,x}{31} + \frac{8}{31}\}$$

```
> plots[implicitplot3d]({3*x - y + 2*z = 4, -7*x + 8*y - z/2 = 1},
> x=-4..4, y=-4..4, z=-4..4, axes=BOXED, orientation=[15, 50]);
```



Podemos comprobar que existe un número infinito de soluciones (esto se debe a que tenemos más variables que ecuaciones). Por otro lado, en la gráfica podemos ver que las funciones implícitas son planos, por lo que un número infinito de soluciones debe corresponder al hecho de que dos planos se intersectan a lo largo de una recta (dado que en este caso no son paralelos).

Intentaremos conseguir una mejor visualización con la superposición de varias estructuras gráficas, por medio de la función **display3d**.

Nos conviene que los planos definidos por las ecuaciones aparezcan sin ***"parches"***, en color solido (por ejemplo amarillo o verde). Asimismo, la recta en la cual se intersectan los planos podra ir superpuesta a estos. La ecuación que describe a dicha recta la obtuvimos al aplicar **solve**, y viene dada en forma paramétrica con parámetro **x**.

Primero calculamos la gráfica de la curva: $[x,\ \frac{25\,x}{31} + \frac{8}{31},\ -\frac{34\,x}{31} + \frac{66}{31}]$, y la asignamos a una variable. Utilizaremos para ello la función **spacecurve** del paquete **plots**, la cual nos permite graficar una curva en el espacio:

```
> recta_int := plots[spacecurve]([x, (25/31)*x + 8/31, -(34/31)*x +
> 66/31], x=-3..-2, axes=boxed, thickness=2, color=red):
```

A continuación, calculamos la gráfica de: $\frac{4-3\,x+y}{2}$.

Para ello podemos usar **plot3d**, ya que se trata de una función dada en forma explícita:

```
> plano_1 := plot3d((4 - 3*x + y)/2, x=-3..-2, y=-2..-1,
> view=-5..20, style=patchnogrid, axes=boxed, color=yellow):
```

Ahora, calculamos la gráfica de: $2(-7x+8y-1)$, también con **plot3d**:

```
> plano_2 := plot3d(2*(-7*x + 8*y - 1), x=-3..-2, y=-2..-1,
> view=-5..20, style=patchnogrid, axes=boxed, color=cyan):
```

Y a continuación desplegamos estas gráficas superpuestas:

```
> plots[display3d]({recta_int, plano_1, plano_2});
```



Con lo cual obtenemos una mejor vista de la intersección.

7.3. Tres ecuaciones con tres incognitas

Si al sistema de la sección anterior le agregamos una ecuación más (que no sea multiplo de alguna de las existentes, es decir, debe haber independencia lineal), obtenemos una solución única del sistema:

```
> solve({3*x - y + 2*z = 4, -7*x + 8*y - z/2 = 1, x + y + z = 1/2},
>   {x, y, z});
```

$$\{y = \frac{-83}{44},\, z = \frac{111}{22},\, x = \frac{-117}{44}\}$$

Graficamente, esto es consistente con el hecho de que tres planos (no paralelos) se intersectan en un punto, si es que dicha intersección existe. Al igual que en el caso anterior, una sola gráfica no muestra claramente esta intersección en un punto:

```
> plots[implicitplot3d]({3*x - y + 2*z = 4, -7*x + 8*y - z/2 = 1,
> x + y + z = 1/2}, x=-4..4, y=-4..4, z=-4..4, axes=BOXED, style=patch);
```



Para trazar mediante **display3d** la superposicion de los planos y el punto de interseccion, definimos la gráfica de cada uno de éstos:

```
> plano_1 := plot3d((4 - 3*x + y)/2, x=-3..-2, y=-2..-1,
> view=-5..20, style=patchnogrid, axes=boxed, color=yellow):

> plano_2 := plot3d(2*(-7*x + 8*y - 1), x=-3..-2, y=-2..-1,
> view=-5..20, style=patchnogrid, axes=boxed, color=cyan):

> plano_3 := plot3d(-x - y + 1/2, x=-3..-2, y=-2..-1, view=-5..20,
> style=patchnogrid, axes=boxed, color=green):
```

A continuación, generamos la gráfica del punto de intersección:

```
> punto_int := plots[pointplot3d]({[-117/44, -83/44, 111/22]}, axes=BOXED,
> style=point, symbol=CIRCLE, symbolsize=25, color=black):
```

Y finalmente desplegamos todas estas estructuras en una misma gráfica:

```
> plots[display3d]({plano_1, plano_2, plano_3, punto_int});
```



De esta forma podemos apreciar el punto en el cual tiene lugar la intersección de los planos.